

Révision introduction au Tableur

4) Et pour quelques fonctions de plus...

1	Objectifs du module	2
2	Les fonctions d'arrondi	2
2.1	<i>L'arrondi d'affichage ou l'affichage tronqué</i>	<i>2</i>
2.2	Syntaxe	3
2.2.1	<i>Fonctions d'arrondi à la décimale</i>	<i>3</i>
2.2.2	<i>Fonctions d'arrondi à un multiple</i>	<i>3</i>
2.3	Définition	4
2.4	Exemples.....	4
3	Les fonctions MIN(), MAX(), MOYENNE()	5
3.1	Syntaxe	5
3.2	Exemples.....	5
4	Les dates	6
4.1	<i>La date du jour et ses formats : syntaxe</i>	<i>7</i>
4.2	<i>Quelques fonctions intéressantes (dates).....</i>	<i>8</i>
5	Pour aller plus loin... Les heures	9
5.1	<i>Cas 1 : le format des heures (heures travaillées par Jean et Julie).....</i>	<i>9</i>
5.2	<i>Cas 2 : horaire pour un examen oral</i>	<i>10</i>
5.3	<i>Cas 3 : décompte des heures</i>	<i>10</i>
6	La fonction MAINTENANT() et le format jj.mm.aa hh:mm:ss.....	11

1 Objectifs du module

Je serai capable...

- de saisir et de modifier les fonctions d'arrondi, de moyenne, minimum, maximum
- de manipuler des dates et des heures

2 Les fonctions d'arrondi

2.1 L'ARRONDI D'AFFICHAGE OU L'AFFICHAGE TRONQUÉ

Il ne faut pas confondre possibilités d'afficher un nombre selon un format donné et fonctions d'arrondi...

Dans ce premier tableau,

Nombre originel	Format d'affichage	Affichage Nombre	Valeur réelle en mémoire
1564.230785	Comptabilité sans décimale affichée	CHF 1 564	1564.230785
0.12345	Comptabilité sans décimale affichée	CHF 0	0.12345
0.123456	Pourcentage	12.35%	0.123456
1564.230785	Monétaire avec deux décimales affichées	CHF 1 564.23	1564.230785

vous remarquez qu'Excel, tout en gardant en mémoire une valeur réelle correspondant au nombre originel, vous permet d'afficher ce nombre de diverses manières : séparateur de milliers, indication de la monnaie, pourcentage, nombre de chiffres après la virgule variables. N'oubliez jamais que la valeur réelle conservée en mémoire, **et elle seule**, sera utilisée dans les formules de la feuille de calcul quel que soit le format d'affichage choisi par l'utilisateur.

Pour tester cette problématique, ouvrez le document [Revision4_Demo_d.xlsx](#) et complétez la feuille "**arrondi.0**" pour obtenir le résultat suivant :

B	C	D
	sans format d'affichage	avec un format d'affichage Nombre sans décimale
Premier nombre	25.6	26
Deuxième nombre	25.6	26
Somme des nombres	51.2	51

Utilisez la fonction SOMME() pour faire la somme entre le premier et le deuxième nombre dans les deux colonnes C et D. Dans la colonne "sans format d'affichage" ne modifiez pas le format ! Dans la colonne "avec un format d'affichage...", appliquez le bon format !

Vous devriez constater les éléments suivants :

1. en supprimant les décimales dans la colonne D, les valeurs affichées sont passées de 25.6 à 26 pour les deux nombres
2. la valeur affichée de la somme est passée de 51.2 à 51
3. les vraies valeurs en mémoire n'ont, elles, pas changées... Si c'était le cas, la somme affichée dans la colonne D devrait être de 52 !

Question : pouvons-nous forcer Excel à retenir la valeur de 26 au lieu de celle de 25.6 ? Nous allons compléter le tableau page suivante en ajoutant une colonne :

	A	B	C	D	E
1					
2			sans format d'affichage	avec un format d'affichage Nombre sans décimale	avec une fonction d'arrondi à l'entier le plus proche
3		Premier nombre	25.6	26	=ARRONDI(D3;0)
4		Deuxième nombre	25.6	26	ARRONDI(nombre;no_chiffres)
5		Somme des nombres	51.2	51	52

Utilisez la fonction ARRONDI() pour le premier et le deuxième nombre; cette fonction arrondit une **valeur** (premier élément de la fonction) à un **nombre de décimales** (deuxième élément de la fonction – ici 0 pour arrondir à l'entier, sans décimale).

La somme des nombres de la colonne E donne maintenant 52 ! La fonction ARRONDI() a obligé Excel à retenir la valeur 26 pour les deux nombres; leur somme donne bien 52.

Nous pouvons bien sûr combiner l'utilisation d'une fonction d'arrondi et un format d'affichage particulier :

Nombre original	Fonction d'arrondi	Affichage Nombre	Valeur réelle en mémoire
1564.230785	arrondi au franc sans format	1564	1564
1564.230785	arrondi au franc avec format d'affichage Comptabilité sans décimale affichée	CHF 1 564	1564
0.12345	arrondi au franc sans format	0	0
0.12345	arrondi au franc avec format d'affichage Comptabilité sans décimale affichée	CHF -	0
0.123456	arrondi à 4 décimales sans format	0.1235	0.1235
0.123456	arrondi à 4 décimales avec format d'affichage Pourcentage	12.35%	0.1235
1564.230785	arrondi aux 5 centimes les plus proches sans format	1564.25	1564.25
1564.230785	arrondi aux 5 centimes les plus proches avec format d'affichage Monétaire avec deux décimales affichées	CHF 1 564.25	1564.25

2.2 SYNTAXE

Il existe deux types de fonctions d'arrondi.

2.2.1 Fonctions d'arrondi à la décimale

ARRONDI(nombre;no_chiffres) : arrondit un nombre à un nombre de décimale (nb_chiffres)

ARRONDI.INF(nombre;no_chiffres) : arrondit à la valeur inférieure

ARRONDI.SUP(nombre;no_chiffres) : arrondit à la valeur supérieure

2.2.2 Fonctions d'arrondi à un multiple

ARRONDI.AU.MULTIPLE(nombre;multiple) : arrondit un nombre à un multiple

PLANCHER(nombre;précision) : arrondit un nombre au multiple inférieur

PLAFOND(nombre;précision) : arrondit un nombre au multiple supérieur

Une fonction d'arrondi à la décimale peut être utilisée pour obtenir le même résultat qu'une fonction d'arrondi à un multiple :

	A	B	C
1	nombre à arrondir	fonction utilisée	Résultat obtenu
2	4.2567	=ARRONDI.AU.MULTIPLE(A2;0.5)	4.5
3	4.2567	=ARRONDI(A3*2;0)/2	4.5

2.3 DÉFINITION

La fonction **ARRONDI()** accepte les arguments suivants :

- un nombre qui représente une valeur à arrondir
- un nombre de chiffres à utiliser pour l'arrondi : si **no_chiffres** est égal ou supérieur à zéro (0), il indique le nombre de décimales; si **no_chiffres** est inférieur à zéro (0), le nombre est arrondi à gauche de la virgule

Formule	Détail	Résultat
=ARRONDI(156789.2634687;3)	Arrondi au millième	156'789.263
=ARRONDI(156789.2634687;2)	Arrondi au centième	156'789.260
=ARRONDI(156789.2634687;1)	Arrondi au dixième	156'789.300
=ARRONDI(156789.2634687;0)	Arrondi à l'unité	156'789.000
=ARRONDI(156789.2634687;-1)	Arrondi à la dizaine	156'790.000
=ARRONDI(156789.2634687;-2)	Arrondi à la centaine	156'800.000
=ARRONDI(156789.2634687;-3)	Arrondi au millier	157'000.000
...		

*NB : la colonne **Résultat** à un format Nombre avec 3 décimales affichées et séparateur de milliers.*

2.4 EXEMPLES

Nombres	Formules d'arrondi à la décimale	Formules d'arrondi à un multiple
18.2780	18.2800	18.2800
18.2730	18.2700	18.2700
18.3333	18.4000	18.4000
18.3333	18.3000	18.3000
3.7200	4.0000	4.0000
3.7200	3.5000	3.5000
145.3670	145.3500	145.3500
146.3670	100.0000	100.0000

Utilisez la feuille **arrondi.1** du classeur **s** pour tester les exemples ci-dessous.

	A	B	C
1	Nombres	Formules d'arrondi à la décimale	Formules d'arrondi à un multiple
2	18.278	=ARRONDI(A2;2)	=ARRONDI.AU.MULTIPLE(A2;0.01)
3	18.273	=ARRONDI(A3;2)	=ARRONDI.AU.MULTIPLE(A3;0.01)
4	18.3333333333333	=ARRONDI.SUP(A4;1)	=PLAFOND(A4;0.1)
5	18.3333333333333	=ARRONDI.INF(A5;1)	=PLANCHER(A5;0.1)
6	3.72	=ARRONDI(A6;0)	=ARRONDI.AU.MULTIPLE(A6;1)
7	3.72	=ARRONDI(A7*2;0)/2	=ARRONDI.AU.MULTIPLE(A7;0.5)
8	145.367	=ARRONDI(A8*20;0)/20	=ARRONDI.AU.MULTIPLE(A8;0.05)
9	146.367	=ARRONDI(A9;-2)	=ARRONDI.AU.MULTIPLE(A9;100)

Après avoir observé attentivement les exemples ci-dessus, faite l'exercice dans la feuille **arrondi.2** du classeur **Revision4_Demo_d.xlsx**.

3 Les fonctions MIN(), MAX(), MOYENNE()

3.1 SYNTAXE

MIN(nombre1;nombre2,...) ou MIN(intervalle) : donne le plus petit de ses arguments

MAX(nombre1 ;nombre2,...) ou MAX(intervalle) : donne le plus grand de ses arguments

MOYENNE(nombre1 ;nombre2 ;...) ou MOYENNE(intervalle) : donne la moyenne arithmétique d'une série de nombres

3.2 EXEMPLES

Minimum :	2.000	Moyenne :	55.475
Maximum :	97.000		
Nombres			
84	48	47	70
45	10	93	87
56	2	97	59
59	27	77	35
97	3	81	43
44	25	2	91
6	89	67	46
54	47	92	87
42	80	78	93
28	31	6	91
Min :	6	2	2
Max :	97	89	97
Moyenne :	51.5	36.2	64

Testez les exemples ci-dessous à partir de la feuille [MinMaxMoyenne.1](#) du classeur [Revision4_Demo_d.xlsx](#).

fx =MIN(B5:E14)				
A	B	C	D	E
	Minimum :	=MIN(B5:E14)		55.475
	Maximum :			
	84		47	70
	45		93	87

fx =MAX(B5:E14)				
A	B	C	D	E
	Minimum :	2.000	Moyenne :	55.475
	Maximum :	=MAX(B5:E14)		
	84		47	70
	45		93	87

fx =MOYENNE(B5:E14)					
A	B	C	D	E	F
	Minimum :	2.000	Moyenne :	=MOYENNE(B5:E14)	
	Maximum :	97.000			
	Nombres				
	84	48	47		
	45	10	93		

Saisissez les formules adéquates dans les lignes 16, 17 et 18 pour les colonnes B à D

Une fois les moyennes de la ligne 18 calculées, modifiez les formules pour arrondir les moyennes à l'unité la plus proche :

Minimum :	2.000	Moyenne :	55.475
Maximum :	97.000		
Nombres			
84	48	47	70
45	10	93	87
56	2	97	59
59	27	77	35
97	3	81	43
44	25	2	91
6	89	67	46
54	47	92	87
42	80	78	93
28	31	6	91
Min :	6	2	35
Max :	97	89	93
Moyenne :	52	36	70

Faite l'exercice dans la feuille **MinMaxMoyenne.2** du classeur **Revision4_Demo_d.xlsx** en vous basant sur la feuille **MinMaxMoyenne.Modele**. Saisissez les formules dans la colonne L, une fois la mise en page/formatage effectué.

4 Les dates

Dans Excel, la date de référence est le 1er janvier 1900 !

Cette date correspond au numéro de série 1 :

date	numéro de série	nb de jours depuis...
01.01.1900	1	1 jour
02.01.1900	2	2 jours
03.01.1900	3	3 jours
04.01.1900	4	4 jours
05.01.1900	5	5 jours
...
09.11.1989	32821	32821 jours
06.04.2016	42466	42466 jours

Le numéro de série 32821 permet d'afficher la date du 9 novembre 1989 (date de la chute du mur de Berlin). Ce numéro de série représente un nombre de jours depuis le 1er janvier 1900. Il est dès lors très facile de manipuler des dates notamment en ajoutant ou en soustrayant des jours à une date donnée.

Par exemple, je souhaite construire un tableau avec la date des prochains 10 lundi à partir du 21 août 2017... Il suffit de saisir la date dans la première cellule et d'ajouter à chaque date le nombre de jours (7) qui me permet de passer d'un lundi à un autre :


	A	B	C	D	E	F	G	H	I	J	K
1		1	2	3	4	5	6	7	8	9	10
2	Lundis	21.08.17	=B2+7	04.09.17	11.09.17	18.09.17	25.09.17	02.10.17	09.10.17	16.10.17	23.10.17
3	Événements	Evmt 1	Evmt 2	Evmt 3	Evmt 4	Evmt 5	Evmt 6	Evmt 7	Evmt 8	Evmt 9	Evmt 10

4.1 LA DATE DU JOUR ET SES FORMATS : SYNTAXE

AUJOURD'HUI() : donne la date du jour automatiquement

Lorsque la fonction AUJOURD'HUI() est saisie dans une cellule non formatée, Excel affiche la date du jour dans le format Date jour.mois.année : par exemple, le 6 avril 2016 sera noté 06.04.16. Si vous appliquez le format Nombre Standard, le numéro de série 42466 s'affiche. La fonction AUJOURD'HUI() vous renverra bien sûr la date du jour où vous faites cet exercice... Le numéro de série correspondant sera bien évidemment plus élevé que celui du 6 avril 2016.

Ces numéros de séries peuvent être personnalisés en utilisant la catégorie **Personnalisée** du format **Nombre** :



Dans le champ "Type :" saisissez jj.mm.aa pour obtenir le format Date standard 06.04.16.

La lettre "j" est utilisée pour formater le jour de la date, la lettre "m" pour le mois de la date et la lettre "a" pour l'année.

Un seul "j" affichera uniquement "6"; deux "jj" afficheront "06"; trois "jjj" afficheront "mer" pour le jour de la semaine abrégé de la date en question (mercredi) et quatre "jjjj" afficheront "mercredi".

"m", "mm", "mmm" et "mmm" affichent respectivement "4", "04", "avr" et "avril".

"a" et "aa" afficheront 16 pour l'année de la date et "aaa" et "aaaa" afficheront 2016.

Pour construire le format d'une date personnalisée, il suffit de combiner les "j", les "m" et les "a" avec des séparateurs : un espace, un point (.), une virgule (,) un tiret (-) ou une barre oblique (/).

Vous pouvez ajouter du texte en plus des séparateurs qui sera saisi entre guillemets :

"le" jjjj j mmmm aaaa permettra d'afficher **le mercredi 6 avril 2016**.

Utilisez la feuille **Date.1** du classeur **Revision4_Demo_d.xlsx** et appliquez les formats date proposés en utilisant la boîte de dialogue des formats Nombre.

4.2 QUELQUES FONCTIONS INTÉRESSANTES (DATES)

Utilisez la feuille **Date.2** du classeur **Revision4_Demo_d.xlsx** et testez les formules ci-après à partir des 2 dates qui y figurent :

JOUR(numéro_de_série) : renvoie le jour du mois de la date (nombre entier entre 1 et 31)

MOIS(numéro_de_série) : renvoie le mois de la date (nombre entier entre 1 -janvier- et 12 -décembre-)

ANNEE(numéro_de_série) : renvoie l'année de la date (nombre entier entre 1900 et 9999)

DATE(année;mois;jour) : renvoie le numéro de série d'une date

JOUR360(date_début;date_fin) : renvoie le nombre de jour entre deux dates en partant de l'hypothèse que chaque mois compte 30 jours (très utile pour le calcul d'intérêts !)

Dans certaines situations, il peut être utile d'accéder au jour, au mois et/ou à l'année d'une date donnée.

Résultat à obtenir :

	date 1	date 2
	18.01.16	25.09.16
Jour	18	25
Mois	1	9
Année	2016	2016
Numéros de série à partir des éléments calculés dans les cellules C5 à D7		
	42387	42638

Pour le nombre de jours exact entre date 1 et date 2, utilisez la fonction DATEDIF ci-dessous :

DATEDIF(date_début;date_fin;unité) : renvoie une différence qui dépend de l'unité choisie...

"y" : renverra une différence d'années, "m" une différence de mois et "d" une différence de jours.

Nombre de jours pour le calcul d'intérêts simples entre date 1 et date 2 :		247
Nombre de jours exact entre date 1 et date 2 :		251

Pour finir, calculez votre âge exact en termes d'années, de mois et de jours à la date d'aujourd'hui :

- pour calculer la différence d'année, utilisez l'unité "y";
- pour calculer la différence de mois, utilisez l'unité "ym" (nous indiquons ici à Excel que nous souhaitons obtenir la différence de mois une fois la différence d'années calculée);
- pour calculer la différence de jour, utilisez l'unité "md" (nous indiquons ici à Excel que nous souhaitons obtenir la différence de jours une fois la différence de mois calculée).

Pour le Dalaï-Lama, son âge exact le 9 août 2016 est de :

La date d'aujourd'hui :	09.08.16
Votre date de naissance :	06.07.35
81 ans,	
1 mois et	
3 jours	

5 Pour aller plus loin... Les heures

Utilisez la feuille **Heure.1** du classeur **Revision4_Demo_d.xlsx** pour découvrir quelques cas d'utilisation des heures dans un tableur.

5.1 CAS 1 : LE FORMAT DES HEURES (HEURES TRAVAILLÉES PAR JEAN ET JULIE)

Employés	Heures (hh:mm)	Numéro de série (standard)	Heures ([h]:mm)	Heures (décimal) (Heures X 24)
Jean	12:45	0.53125	12:45	12.75
Julie	21:32	0.897222222	21:32	21.53333333
Total	10:17	1.428472222	34:17	34.28333333

Saisissez dans les cellules C4 et C5 les heures 12:45 et 21:32 (insérez les 2 points entre les chiffres pour indiquer à Excel que vous manipulez des heures). Dans la cellule C6, faites la somme des heures travaillées par Jean et Julie (utilisez la fonction SOMME()). Ensuite, sélectionnez les cellules C4 à C6 et accédez au format de nombre personnalisé et insérez dans le champ Type le format **hh:mm**.

Saisissez ensuite les formules ci-dessous (utilisez le \$ et le copier-coller pour aller plus vite) :

Numéro de série (standard)	Heures ([h]:mm)	Heures (décimal) (Heures X 24)
=C4	=C4	=C4*24
=C5	=C5	=C5*24
=C6	=C6	=C6*24

Les colonnes D et E reprennent simplement les valeurs de la colonne C.

Appliquez le format de nombre Standard pour la colonne D et F.

Appliquez le format de nombre [h]:mm pour la colonne E.

Remarques :

- l'affichage du total des heures dans la colonne C et la colonne E est différent... Dans la colonne C, le total entre 12:45 de travail et 21:32 donne **10:17**, ce qui au premier abord est faux; en fait le format hh:mm ne permet pas d'afficher des heures au-delà de 24h ! 10:17 représente 10h17 de plus que les 24h. Dès le moment où l'addition d'heures dépasse 24h, veillez à insérer les "[]" dans le format ([h]:mm), ce qui a été fait dans la colonne E.
- dans la colonne D, Excel affiche le numéro de série correspondant à l'heure saisie. C'est une part d'une journée de 24h, qui mis en forme avec le format heure permet d'afficher les heures correctement.
- dans la colonne F, vous avez multiplié le numéro de série par 24 pour retrouver le nombre d'heure en format décimal (attention de bien utiliser le format standard ou nombre pour afficher correctement le résultat); ce nombre en format décimal pourra par exemple être multiplié par le salaire horaire des employés pour déterminer le salaire qui leur sera versé.

5.2 CAS 2 : HORAIRE POUR UN EXAMEN ORAL

	début examen	08:00	
	Durée préparation	20 minutes	
	Durée examen	20 minutes	
	Durée discussion	10 minutes	
Elèves	Préparation	Examen oral	Discussion Note
Jean	08:00	08:20	08:40
Julie	08:50	09:10	09:30
Joanne	09:40	10:00	10:20
Jasmine	10:30	10:50	11:10

Saisissez dans les cellules D9 à D12 les heures, respectivement, 08:00, 00:20, 00:20 et 00:10.

Appliquez le format personnalisé **[m] "minutes"** pour les cellules D10 à D12 pour obtenir un affichage en minutes (ne pas mettre les crochets autour du m affichera la valeur 1 – et correspondra à un mois).

Si vous deviez travailler avec des heures uniquement, utilisez **[h] "heures"** comme format et si vous deviez travailler avec des secondes, utilisez **[s] "secondes"**.

Insérez ensuite les formules nécessaires au calcul des heures de passage des élèves (utilisez les \$ et le copier-coller) :

Elèves	Préparation	Examen oral	Discussion Note
Jean	=D9	=C15+\$D\$10	=D15+\$D\$11
Julie	=E15+\$D\$12	=C16+\$D\$10	=D16+\$D\$11
Joanne	=E16+\$D\$12	=C17+\$D\$10	=D17+\$D\$11
Jasmine	=E17+\$D\$12	=C18+\$D\$10	=D18+\$D\$11

5.3 CAS 3 : DÉCOMPTE DES HEURES

Semaine du lundi 8 août					
au vendredi 12 août					
	Matin arr.	Matin dép.	Après-midi arr.	Après-midi dép.	Heures travaillées
Lundi	7:30	11:30	13:34	17:00	07:26
Mardi	7:38	12:30	14:00	16:55	07:47
Mercredi	7:15	12:04	14:00	18:10	08:59
Jeudi	7:23	12:08	13:15	19:30	11:00
Vendredi	7:00	11:45	13:10	16:30	08:05
				Total	43:17
				Pour le calcul du salaire	43.28333333

Dans la colonne G, insérez les formules qui permettent de calculer les heures travaillées d'un employé pour une semaine donnée (d'abord en format heures et ensuite en format décimal pour le calcul du salaire).

Appliquez le bon format pour les cellules C21 et C22 !

6 La fonction MAINTENANT() et le format jj.mm.aa hh:mm:ss

Utilisez la feuille **Maintenant** du classeur **Revision4_Demo_d.xlsx**.

	MAINTENANT()	Date du jour	Heure
Date et heure de la facture	13.08.16 11:21:29	13.08.16	11:21:29

Cette fonction vous permet de générer la date et l'heure dans la même cellule. Le format de base proposé par Excel est **jj.mm.aa hh:mm**. Dans la cellule C3, insérez la fonction MAINTENANT(). Puis appliquez le format personnalisé **jj.mm.aa hh:mm:ss** (attention de placer un espace entre le format date et le format heure).

Vous pouvez vous aussi saisir à la main (sans utiliser la fonction maintenant) la date et l'heure dans la même cellule. Essayez de saisir dans la cellule C5 la date et l'heure : **22.09.16 14:23:45**. Excel vous propose à nouveau le même format que lors de l'utilisation de la fonction MAINTENANT(); il ne vous affiche pas les secondes. Appliquez le même format que pour la cellule C3, avec les secondes.

Dans la cellule C6, saisissez simplement la formule **=C5**. Puis appliquez le format **Standard**.

le 22 septembre à 14h23	22.09.16 14:23:45	22.09.16	14:23:45
	42635.59983	42635	0.59982639

Cette date du 22 septembre à 14h23 génère un numéro de série comprenant deux parties :

- la première partie, le nombre entier 42635 représente le numéro de série de la date;
- la deuxième partie, les chiffres après la virgule (0.59983...) représente la fraction d'un jour en heures.

Il est donc facile, à partir de là, de séparer la date des heures dans deux cellules différentes...

Dans la cellule D3, utilisez la fonction ENT() (**=ENT(C3)**) qui permet de prélever la partie entière d'un nombre (ou de notre numéro de série). Puis copiez cette formule (et uniquement la formule !) dans les cellules D5 et D6.

Pour obtenir l'heure, il suffit de soustraire au numéro de série complet la partie entière que vous venez de récupérer... Saisissez la formule **=C3-D3** dans la cellule E3. Puis copiez cette formule dans les cellule E5 et E6. Réajustez les formats si nécessaire.